Exercise 1

a) Find the tangent(s) to the curve x=t^2+2t, y=t^4-4t^2 at the point (0,0). b) Find the point with vertical tangent.

a) The curve and plot

```
> x:=t->t^2+2*t;
  y:=t->t^4-4*t^2;
  plot([x(t),y(t),t=-3..1]);
```

$$x := t \to t^2 + 2\,t$$

$$y := t \to t^4 - 4\,t^2$$



Values of t at (0,0). They are the common solutions to the equations x(t)=0 and y(t)=0

```
> solve({x(t)=0,y(t)=0},{t});
```

(1)

$$\{t=0\},\ \{t=-2\} \tag{1}$$

The derivative of the position (velocity) at t points in the direction of the tangent at t

```
> dx:=D(x);
  dy:=D(y);
```

$$dx := t \rightarrow 2\,t+2 \tag{2}$$
$$dy := t \rightarrow 4\,t^3-8\,t$$

Tangent line at t1=0

```
> xt1:=x(0)+dx(0)*t;
  yt1:=y(0)+dy(0)*t;
```

$$xt1 := 2\,t \tag{3}$$
$$yt1 := 0$$

Tangent line at t2=-2

```
> xt2:=x(-2)+dx(-2)*t;
  yt2:=y(-2)+dy(-2)*t;
```

$$xt2 := -2\,t \tag{4}$$
$$yt2 := -16\,t$$

Plot to check that these are the tangents at (0,0)

```
> plot([[x(t),y(t),t=-3..1],[xt1,yt1,t=-1..1],[xt2,yt2,t=-1..1]],
  axes=none);
```



b) At Points of vertical tangent the horizontal component of the velocity should be zero, i.e. x´(t)=0 and the vertical component y´(t) not zero

```
> t3:=solve(dx(t)=0,t);
```

$$t3 := -1 \qquad (5)$$

At this point y'(-1) is not zero

```
> dy(t3);
```

$$4 \qquad (6)$$

There is a vertical tangent at the point

```
> x(t3);
  y(t3);
```

$$-1 \qquad (7)$$

$$-3$$

Equation of the vertical tangent line

```
> xtv:=x(t3)+dx(t3)*t;
  ytv:=y(t3)+dy(t3)*t;
```

$$xtv := -1 \qquad \textbf{(8)}$$

$$ytv := -3 + 4\,t$$

Plot

```
> plot([[x(t),y(t),t=-3..1],[xtv,ytv,t=-2..2]]);
```



Exercise 2
Given the function f(x,y)=x^2+xy; a) Find the directional derivative of f at point (1,1) in the direction **d**= 2**i**+**j**; b) in what direction(s) is the derivative of f at P equal to zero?

a) The function

```
> f:=(x,y)->x^2+x*y;
```

$$f := (x, y) \to x^2 + x\,y \qquad \textbf{(9)}$$

Partial derivatives of f

```
> fx:=D[1](f);
  fy:=D[2](f);
```

$$fx := (x, y) \to 2\,x + y \qquad \textbf{(10)}$$

$$fy := (x, y) \to x$$

Components of the gradient vector of f at P(1,1)

```
> gx:=fx(1,1);
  gy:=fy(1,1);
```

$$gx := 3 \qquad \textbf{(11)}$$

$$gy := 1$$

Unit vector in the direction **d**= 2**i**+**j**

```
> dx:=2;
  dy:=1;
  dnorm:=sqrt(dx^2+dy^2);
  ux:=dx/dnorm;
  uy:=dy/dnorm;
```

$$dx := 2 \qquad \textbf{(12)}$$

$$dy := 1$$

$$dnorm := \sqrt{5}$$

$$ux := \frac{2\sqrt{5}}{5}$$

$$uy := \frac{\sqrt{5}}{5}$$

The derivative or rate of change of f at P(1,1) in the direction **u** is the dot product of the gradient vector at P and the unit vector **u**

```
> Duf:=gx*ux+gy*uy;
```

$$Duf := \frac{7\sqrt{5}}{5} \qquad \textbf{(13)}$$

b) The unit vectors orthogonal to the gradient at P are **n**=(gy**i**-gx**j**)/gnorm or **n**=(-gy**i**+gx**j**)/gnorm where gnorm is the norm of the gradient**.** Since the gradient is perpendicular to the level curve of f that goes trough the point P these vectors give the direction of the tangent line to the level curve at P.

```
> gnorm:=sqrt(gx^2+gy^2);
  nx:=gy/gnorm;
  ny:=-gx/gnorm;
```

$$gnorm := \sqrt{10} \qquad \textbf{(14)}$$

$$nx := \frac{\sqrt{10}}{10}$$

$$ny := -\frac{3\sqrt{10}}{10}$$

or

```
> -nx;
  -ny;
```

$$-\frac{\sqrt{10}}{10} \quad (15)$$

$$\frac{3\sqrt{10}}{10}$$

In these directions the rate of change of f at P is zero.

```
> Dfn:=gx*nx+gy*ny;
```

$$Dfn := 0 \quad (16)$$

Exercise 3
*a)Find the maximum and minimum values of f(x,y)=x^2+y^2-2*x-2*y+3 in the region D bounded by the lines x=0, y=0 and x+y=3.*
b) Evaluate the integral of f (x,y) over D.

a) Function f and plot of f over D

```
> restart:
  f:=(x,y)->x^2+y^2-2*x-2*y+3;
```

$$f := (x, y) \rightarrow x^2 + y^2 - 2x - 2y + 3 \quad (17)$$

```
> plot3d({f(x,y),0},x=0..3,y=0..3-x,style=patchcontour,axes=normal)
  ;
```



Partial derivatives

```
> fx:=D[1](f);
  fy:=D[2](f);
```

$$fx := (x, y) \to 2\,x - 2 \quad (18)$$

$$fy := (x, y) \to 2\,y - 2$$

Critical points

```
> solve({fx(x,y)=0,fy(x,y)=0},{x,y});
```

$$\{x = 1, y = 1\} \quad (19)$$

Level curves

```
> with(plots):
  contourplot(f(x,y),x=0..3,y=0..3-x);
```



Looking at the level curves and the corresponding contours in the surface we observe that the function reaches an absolute minimum at the critical point (1,1) and then it increases in all directions more and more rapidly. Therefore the absolute maximum values of f over D will be reached at the points (3,0) and (0,3) which are the farthest points away from point (1,1). The minimum and maximum values are

```
> fmin:=f(1,1);
  fmax:=f(3,0);
  fmax:=f(0,3);
```

$$fmin := 1 \quad \textbf{(20)}$$

$$fmax := 6$$

$$fmax := 6$$

b) To evaluate the integral of f over D first we integrate with respect to y from axis y=0 to the line y=3-x and then with respect to x from x=0 to x=3 to get

```
> V:=Int(f(x,y),[y=0..3-x,x=0..3]);
  value(V);
```

$$V := \int_0^3 \int_0^{3-x} \left(x^2 + y^2 - 2x - 2y + 3\right) \, dy \, dx \quad (21)$$

$$9$$